
$oninline 
/* 
 GAMS program used to estimate technical efficiency,  using a directional distance function model.  
The directional distance function seeks to simultaneously expand outputs and contract inputs 
 
 
 Author:  John B. Walden 
          NMFS/NEFSC 
          166 Water St.  
          Woods Hole, MA  02543 
          (508) 495-2355 
          John.Walden@Noaa.Gov 
 
This version will estimate technical efficiency for each DMU, using directional distance functions.  
 
Fare, R. and S. Grosskopf. 2000. "Theory and Application of Directional Distance Functions." Journal of Productivity 
Analysis,13:93-103. 
 
*/ 
 
/* The following line turns off listing of some elements in the GAMS listing file*/ 
 
 
$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF 
 
 
OPTION SOLPRINT=OFF, SYSOUT=OFF, LIMROW=0, LIMCOL=0; 
 
 
/*NEXT DEFINE INPUTS AND OUTPUTS*/ 
 
 
SET INOUT /out1*out6, fix1*fix3, var1, var2/ 
    OUTPUT(INOUT) /out1*out6/ 
    INPUT(INOUT) /fix1*fix3, var1, var2/ 
 
 
    OBS /1*200/ 
    SUBOBS(OBS) /1*82/ 
    ACTOBS(OBS) 
 
/*We have allocated enough memory for 200 observations, but ourdata set only contains 82 observations (subobs)*/ 
; 
 
/*Next, define an alias for the set SUBOBS */ 
 
alias (subobs, subobs1) 
 
/* The include statement below reads in an external datafile which contains a table of observations, inputs 
and outputs. The files is in comma separted value format (CSV) which can be written from, and read 
into spreadsheet programs 
 
The offlisting command means that the data won't be included in the listing file. 
 
*/ 
 



$OFFLISTING 
 
TABLE ACT(OBS,INOUT)  INPUT OUTPUT TABLE 
$ondelim 
$INCLUDE "cap1.csv" 
$offdelim 
 
$ONLISTING 
 
 PARAMETER 
 
   score1(obs)  efficiency scores 
; 
 
 
VARIABLES 
 
 Beta objective value 
 weight(obs)  weights; 
 
 POSITIVE Variable weight; 
 
 EQUATIONS 
 
   CONSTR1(OUTPUT, OBS)  DEA constraint for each output 
   CONSTR2(INPUT,OBS) DEA Constraint for calculating TE; 
 
   CONSTR1(OUTPUT, ACTOBS).. SUM(SUBOBS,WEIGHT(SUBOBS)*ACT(SUBOBS,OUTPUT)) 
                                =G= (1+beta)*ACT(ACTOBS, OUTPUT); 
 
   CONSTR2(INPUT, ACTOBS).. SUM(SUBOBS,WEIGHT(SUBOBS)*ACT(SUBOBS,INPUT))  
                                 =L=(1-Beta)*ACT(ACTOBS,INPUT); 
 
 
/*Define a parameter to hold results for each pass throughthe loop*/ 
 
 
/*Define an external file to hold results which tell whether model solved 
at each iteration*/ 
 
file primal /te_res.txt/; 
 
 MODEL TE /CONSTR1, CONSTR2/; 
 
 
te.solprint=2;           /*Turn off writing results to solution file*/ 
te.solvelink=2;       /*Keep model in memory. Improves solution time*/ 
 
 LOOP(SUBOBS1, 
 
 
      ACTOBS(OBS)=NO; 
      ACTOBS(SUBOBS1)=YES; 
 
      SOLVE TE maximizing Beta USING LP; 
 



      score1(subobs1)=Beta.l; 
 
     put primal; 
 
      if ((te.modelstat eq 1 and  te.solvestat eq 1), 
 
        put @1, subobs1.tl, @10, "optimal", @20, "normal completion" / 
 
      else 
 
        put @1, subobs1.tl, @10, te.modelstat:>2:0, 
             @20, te.solvestat:>2:0/ 
      ); 
 
   ); 
 
 
/*The next file is to output results to a file to be imported 
into a spreadsheet program.  Results could also be printed to 
the listing file with the use of the display command*/ 
 
 
file res /"dd_te_crs.csv"/ ; 
 
res.pc=5; 
res.pw=200; 
 
put res; 
 
put 'Obs', 'Beta', 
  
  
put / 
 
 loop (subobs1, 
 
    put / 
    put subobs1.tl, score1(subobs1), 
 
 ); 
putclose; 
 
 


